

Descrierea surselor independente în programul PSPICE

Sursele independente sunt de două tipuri:

- surse independente de tensiune, notate cu V;
- surse independente de curent, notate cu I.

Formele generale de descriere a celor două tipuri de surse sunt similare, singura diferență fiind caracterul folosit pentru indicarea tipului sursei:

V<nume> <N+> <N-> [[DC] <valoare>] [AC <modul> [fază]]
+[specificație_regim_tranzitoriu]

Exemple:


Vcc 22 0 15

Vin 1 0 DC 3.5 AC 1

Vsemnal 3 0 AC 1 90

ISIN 0 1 DC 100m SIN(100m 10m 1k)

Vin 1 0 AC 1 PULSE(0 5 1 2 2 5 15)



În forma generală de descriere, N+ și N- reprezintă nodul pozitiv, respectiv negativ al sursei. Pentru sursa independentă de tensiune, curentul pozitiv circulă prin sursă de la nodul N- la nodul N+. În cazul sursei independente de curent, acesta circulă prin sursă de la nodul N+ la nodul N-.

Sursele independente dispun de următoarele specificații opționale:

- [[DC] <valoare>] pentru specificarea valorii componentei continue a sursei. Această valoare este luată în considerare la calculul punctului static de funcționare, la analiza de curent continuu și la analiza în domeniul timp (dacă [specificație_regim_tranzitoriu] nu este prezentă). Dacă sursei trebuie să i se specifice numai valoarea componentei continue, opțiunea [DC] poate fi omisă.
- [AC <modul> [fază]] pentru analiza de semnal mic (AC), sursa devenind sursă de semnal (mărime complexă), faza inițială fiind măsurată în grade și având valoarea implicită 0. Deoarece la analiza de semnal mic circuitul este liniarizat în jurul punctului static de funcționare, analiza fiind liniară, se recomandă ca parametrul <modul> să aibă valoarea 1.
- [specificație_regim_tranzitoriu] pentru analiza în domeniul timp. Aceasta poate fi una din următoarele forme de undă:

PULSE(<parametri>) formă de undă pulsatorie;

SIN(<parametri>) formă de undă sinusoidală;

EXP(<parametri>) formă de undă exponențială;

PWL(<parametri>) formă de undă liniară pe porțiuni;

SFFM(<parametri>) formă de undă modulată în frecvență.

Descrierea formelor de undă în domeniul timp

Forma de undă pulsatorie are forma generală de descriere:

PULSE(<v1> <v2> <td> <tr> <tf> <pw> <per>)


Exemple:

V1 5 0 PULSE(-2 3 5m 2m 4m 3m 15m) ; sursa de tensiune pulsatorie

Ipulse 0 4 PULSE(100m 500m 0.5 3 2 5 15) ; sursa de curent pulsatoriu

Parametrii forme de undă au următoarea semnificație:

Parametru	Semnificație	Unitate de măsură	Valoare implicită
v1/i1	Tensiunea/curentul inițial	Volt/Amper	-
v2/i2	Tensiunea/curentul pulsului	Volt/Amper	-
td	Timpul de întârziere	sec	0
tr	Timpul de creștere	sec	TSTEP
tf	Timpul de cădere	sec	TSTEP
pw	Lățimea pulsului	sec	TSTOP
per	Perioada pulsului	sec	TSTOP



Variabilele TSTEP și TSTOP reprezintă valorile implicite ale unor parametri ai forme de undă pulsatorii, stabilite în cadrul analizei .TRAN. Astfel, TSTEP este pasul de tipărire iar TSTOP este timpul final al analizei tranzitorii.


Forma de undă pulsatorie și semnificația parametrilor săi sunt prezentate în Figura 1.

Fig. 1 Forma de undă pulsatorie

Atribuind valori convenabile parametrilor forme de undă pulsatorii, pot fi descrise diverse forme de undă particulare. În continuare sunt prezentate câteva din cele mai importante cazuri particulare.


Forma de undă dreptunghiulară

($tr \ll per$, $tf \ll per$, $tr \neq 0$, $tf \neq 0$)




Forma de undă triunghiulară


($pw \ll per$, $tr + tf = per$, $pw \neq 0$)




Forma “dinte de fierăstrău”
 ($tr \ll per$, $pw \ll per$, $per = tf$)



Forma “treaptă”
 ($tr \ll TSTOP$, $pw > TSTOP$)



Forma “rampă”
 ($tr > TSTOP$)



Forma de undă sinusoidală are următoarea formă generală de descriere:

$$\text{SIN}(\langle V0 \rangle \langle A \rangle \langle f \rangle \langle td \rangle \langle df \rangle \langle faza \rangle)$$

Parametrii și aspectul formei de undă sinusoidale sunt prezentați în continuare:

Parametru	Semnificație	Unitate de măsură	Valoare implicită
V0/I0	Componenta continuă	Volt/Amper	-
A	Amplitudinea	Volt/Amper	-
f	Frecvența	Hertz	1/TSTOP
td	Timpul de întârziere	sec	0
df	Factorul de amortizare	1/sec	0
faza	Faza inițială	grade	0


Forma de undă sinusoidală este descrisă de următoarele expresii:

$$\text{Pentru } 0 < t < td: v(t) = V0 + A \cdot \sin(2\pi \cdot faza/360^\circ);$$

$$\text{Pentru } t > td: v(t) = V0 + A \cdot \sin(2\pi \cdot f \cdot (t - td) + 2\pi \cdot faza/360^\circ) \cdot e^{-(t-td) \cdot df}.$$

Forma de undă din figura de mai jos corespunde unei surse independente de tensiune ce are următoarea descriere:

$$Vsin \ 1 \ 0 \ SIN(2 \ 3 \ 5 \ 0.5 \ 1 \ 30)$$




Forma de undă exponențială are următoarea formă generală de descriere:

EXP(<V1> <V2> <td1> <tc1> <td2> <tc2>)

Exemplu:

Vexp 1 0 EXP(1 5 0.5 0.1 2 0.4)



Parametrii forme de undă exponențiale sunt prezentați în tabelul următor:

Parametru	Semnificație	Unitate de măsură	Valoare implicită
V1/I1	Tensiunea/curentul inițial	Volt/Amper	-
V2/I2	Nivelul spre care tinde forma exponențială	Volt/Amper	-

td1	Timpul la care începe creșterea/scăderea	sec	0
tc1	Constanta de timp pentru creștere/scădere	sec	TSTEP
td2	Timpul la care începe scăderea/creșterea	sec	td1+TSTEP
tc2	Constanta de timp pentru scădere/creștere	sec	TSTEP

Expresiile formei de unda exponențiale sunt:

Pentru $0 < t < td1$: $v(t) = V1$;

Pentru $td1 < t < td2$: $v(t) = V1 + (V2 - V1) \cdot (1 - e^{-(t - td1) / tc1})$;

Pentru $t > td2$: $v(t) = V1 + (V2 - V1) \cdot ((1 - e^{-(td2 - td1) / tc1}) - (1 - e^{-(t - td2) / tc2}))$.

Forma de undă liniară pe porțiuni are următoarea formă generală:


PWL <punct_de_frângere>*

<punct_de_frângere> este de tipul (tn,vn), pentru a specifica coordonatele unui punct de frângere, unde tn reprezintă timpul corespunzător punctului de frângere, iar vn reprezintă valoarea de tensiune/curent corespunzătoare punctului de frângere.

În acest caz, forma de undă liniară pe porțiuni va fi alcătuită din segmente de dreaptă care vor uni punctele de frângere. În afara domeniului punctelor de frângere, valoarea sursei va fi constantă, la nivelul punctelor extreme.

Exemplu:

Vin 1 0 PWL (1,0.5) (2,2) (4,3.5) (6,1) (7,2)



Alte forme generale admise:

REPEAT FOR <n> <punct_de_frângere>* ENDREPEAT

pentru a specifica o formă de undă liniară pe porțiuni, repetată de n ori;

REPEAT FOREVER <puncte_de_frangere>* ENDREPEAT

pentru a specifica o formă de undă liniară pe porțiuni, repetată la infinit.

Exemple:

V1 3 8 PWL REPEAT FOR 3 (1,1) (2,3) (3,3.5) (4,1) ENDREPEAT

V1 3 8 PWL REPEAT FOREVER (1,1) (2,3) (3,3.5) (4,1) ENDREPEAT

Cu ajutorul descrierii PWL (piecewise linear) pot fi descrise diverse forme de undă particulare, prezentate la forma de undă pulsatorie: forma dreptunghiulară, triunghiulară, treaptă, rampă etc.

Forma de undă modulată în frecvență are următoarea formă de descriere:

SFFM(<V0> <A> <FC> <M> <FM>)

Exemplu:


Vsffm 1 0 SFFM(1 2 10k 2.75 2k)

Parametrii formei de undă modulată în frecvență sunt prezentați în tabelul următor:

Parametru	Semnificație	Unitate de măsură	Valoare implicită
V0/I0	Componenta continuă	Volt/Amper	-
A	Amplitudinea tensiunii (curentului)	Volt/Amper	-
FC	Frecvența purtătoare	Hertz	1/TSTOP
M	Indicele de modulație în frecvență	-	0
FM	Frecvența modulatorie	Hertz	1/TSTOP

Variabila TSTOP este specificată în cadrul comenzii .TRAN și reprezintă timpul până la care se efectuează analiza de regim tranzitoriu.

Forma de undă modulată în frecvență corespunzătoare sursei Vsffm din exemplu este prezentată în figura următoare:



Expresia formei de undă modulate în frecvență este:

$$v_{\text{SFFM}}(t) = V_0 + A \cdot \sin(2\pi \cdot FC \cdot t + M \cdot \sin(2\pi \cdot FM \cdot t)).$$

Aplicație:

Să se descrie în PSPICE un circuit care conține sursele de tensiune care furnizează următoarele forme de undă. Circuitul se va simula în domeniul timp și se vor vizualiza formele de undă descrise mai jos. Fișierul de intrare va avea următoarea formă:

*Surse independente

V1 ...

R1 ...

...

.tran 0.5m 15m 0 0.1m

.probe

.end

